

Adatbázisrendszerek II. – 12. Practice

Topic: Oracle PL/SQL programozás, PLSQL 5

Repository: NEPTUNKOD_DB2Gyak

Folder: NEPTUNKOD_0515

Forrás file-k: forrás fájlok .SQL

Töltse fel a GitHub rendszer aktuális mappába a jegyzőkönyvet és forrás fájlokat!

Határidő: aktuális gyakorlat időpontja, ill. módosítás esetén 2025.05.21.

Feladatok

Projekt név: NEPTUNKOD_PLSQL5

Feladat

Először a minta feladatsort készítse el, majd a Zoo táblához készítse el a feladatot.

Zoo						
ID	Nev	Fajta	Született	Suly	Ertek	Gondozo

- Készítsen egy napló táblát a Zoo táblához.
- Készítse el a beszúrást, módosítást, és törlést naplózó triggereket.
 - A napló tábla tartalmazza a naplózott művelet összes adatát, és az adatokból vissza lehessen állítani a módosításokat, és a törölt rekordokat.
- Próbálja ki a naplózás működését.
- Készítsen egy `DeleteAllTriggers()` nevű tárolt eljárást, mely letörli az összes létrehozott triggerét!

5. Készítsen egy `ChangeBackPet(num)` tárolt eljárást, mely egy megadott sorszámú naplóbejegyzés alapján visszaállítja egy módosított állat adatait a módosítás előtti értékekre. Futtassa az eljárást! (+2 pont)
6. Készítsen egy `ResetPet(num)` nevű tárolt eljárást, mely egy megadott sorszámú naplóbejegyzés alapján visszaállít egy törölt állatot a Zoo táblába. Futtassa az eljárást! (+2 pont)

Triggerek

AutoNaplo			Auto						
Datum	Esemeny	Adat	ID	Rendszam	Tipus	Szin	Kor	Ar	Muszaki_erv

- Nagyon sűrűn használják a triggeret DML műveletek naplózására
- Az **AutoNaplo** tábla létrehozási kódja:

```
CREATE TABLE AutoNaplo (  
    Datum timestamp,  
    Esemeny varchar2(30),  
    Adat varchar2(150)  
);
```

Triggerek

AutoNaplo			Auto						
Datum	Esemeny	Adat	ID	Rendszam	Tipus	Szin	Kor	Ar	Muszaki_erv

```
CREATE OR REPLACE Trigger TAutoInsert AFTER INSERT ON Auto FOR EACH ROW  
BEGIN  
    INSERT INTO AutoNaplo VALUES(sysdate , 'Beszúrás', :new.ID||'_'  
        ||:new.Rendszam||'_'||:new.Tipus||'_'||:new.Szin||'_'||:new.Kor  
        ||'_'||:new.Ar||'_'||:new.Muszaki_erv);  
END;
```

```
CREATE OR REPLACE Trigger TAutoUpdate AFTER UPDATE ON Auto FOR EACH ROW  
BEGIN  
    INSERT INTO AutoNaplo VALUES(sysdate , 'Módosítás', :new.ID||'_'  
        ||:new.Rendszam||'_'||:new.Tipus||'_'||:new.Szin||'_'||:new.Kor  
        ||'_'||:new.Ar||'_'||:new.Muszaki_erv);  
END;
```

```
CREATE OR REPLACE Trigger TAutoDelete AFTER DELETE ON Auto FOR EACH ROW  
BEGIN  
    INSERT INTO AutoNaplo VALUES(sysdate , 'Törlés', :old.ID||'_'  
        ||:old.Rendszam||'_'||:old.Tipus||'_'||:old.Szin||'_'||:old.Kor  
        ||'_'||:old.Ar||'_'||:old.Muszaki_erv);  
END;
```

```
BEGIN
  AutoCsomag.UjAuto(13, 'SLK228', 'Mercedes', 'Piros', 6, 7200000, '02.22.2024');
  AutoCsomag.AutoModosit(13, 'Ezüst');
  AutoCsomag.AutoTorol(13);
END;
```

```
SELECT * FROM AutoNaplo;
```

DATUM	ESEMENY	ADAT
13-APR-23 10.15.52.000000 AM	Beszúrás	13_SLK228_Mercedes_Piros_6_7200000_02/22/2024
13-APR-23 10.15.52.000000 AM	Módosítás	13_SLK228_Mercedes_Ezüst_6_7200000_02/22/2024
13-APR-23 10.15.52.000000 AM	Törlés	13_SLK228_Mercedes_Ezüst_6_7200000_02/22/2024

Helyreállítás

AutoNaplo			Auto						
Datum	Esemeny	Adat	ID	Rendszam	Tipus	Szin	Kor	Ar	Muszaki_erv

- Állítsuk vissza a napló táblából a megadott ID-jű törölt adatsort. Logika: az adat értékében megkeressük az aláhúzás karaktereket, kidaraboljuk közülük az értékeket, és beszúrjuk az autó táblába az adatokat.

```
CREATE OR REPLACE Procedure Helyreallit (idbe number) is
  dstr varchar2(150);
  pos1 integer;
  pos2 integer;
  rsz char(6);
  tip char(30);
  szn varchar2(20);
  krr varchar2(3);
  arr varchar2(10);
  msz varchar2(10);
BEGIN
  SELECT ADAT INTO dstr FROM Autonaplo WHERE
  SUBSTR(adat, 1, instr(adat, '_', 1, 1)-1)= TO_CHAR(idbe) AND esemeny='Törlés';
  dstr:= REPLACE(dstr, ' ', '');

```

ADAT
13_SLK228_Mercedes_Ezüst_6_7200000_02/22/2024

Helyreállítás

AutoNaplo			Auto						
Datum	Esemeny	Adat	ID	Rendszam	Típus	Szin	Kor	Ar	Muszaki_erv

```
pos1:= INSTR(dstr, '_', 1, 1);
pos2:= INSTR(dstr, '_', 1, 2);
rsz:= SUBSTR(dstr, pos1+1, pos2-pos1-1);
pos1:= pos2;
pos2:= INSTR(dstr, '_', 1, 3);
tip:= SUBSTR(dstr, pos1+1, pos2-pos1-1);
pos1:= pos2;
pos2:= INSTR(dstr, '_', 1, 4);
szn:= SUBSTR(dstr, pos1+1, pos2-pos1-1);
pos1:= pos2;
pos2:= INSTR(dstr, '_', 1, 5);
krr:= SUBSTR(dstr, pos1+1, pos2-pos1-1);
pos1:= pos2;
pos2:= INSTR(dstr, '_', 1, 6);
arr:= SUBSTR(dstr, pos1+1, pos2-pos1-1);
pos1:= pos2;
msz:= SUBSTR(dstr, pos1+1, 10);
INSERT INTO Auto VALUES(idbe, rsz, tip, szn, krr, arr, msz);
END;
```

Helyreállítás

AutoNaplo			Auto						
Datum	Esemeny	Adat	ID	Rendszam	Típus	Szin	Kor	Ar	Muszaki_erv

- A bemutatott módszer működik, de eléggé körülményes.
- Javaslat: olyan napló táblát készítsen, amibe minden mező külön beszúrható. Picit komplikáltabb beszúrni az adatokat, de visszaállítani sokkal egyszerűbb:

AutoNaplo									
Datum	Esemeny	ID	Rendszam	Típus	Szin	Kor	Ar	Muszaki	

Univerzális Naplózás

AutoNaplo					
Datum	Esemeny	Adat			
Auto					
ID	Rendszam	Tipus	Szin	Kor	Ar
			Muszaki_erv		

```
CREATE OR REPLACE Trigger UniverzalisNaplozas
    AFTER INSERT OR UPDATE OR DELETE ON Auto FOR EACH ROW
BEGIN
    IF inserting THEN INSERT INTO AutoNaplo VALUES(sysdate,'Beszúrás',
        :new.ID||'_'||:new.Rendszam||'_'||:new.Tipus||'_'||:new.Szin
        ||'_'||:new.Kor||'_'||:new.Ar||'_'||:new.Muszaki_erv);

    ELSIF updating THEN INSERT INTO AutoNaplo VALUES(sysdate,'Módosítás',
        :new.ID||'_'||:new.Rendszam||'_'||:new.Tipus||'_'||:new.Szin
        ||'_'||:new.Kor||'_'||:new.Ar||'_'||:new.Muszaki_erv);
    ELSIF deleting THEN INSERT INTO AutoNaplo VALUES(sysdate,'Törlés',
        :old.ID||'_'||:old.Rendszam||'_'||:old.Tipus||'_'||:old.Szin
        ||'_'||:old.Kor||'_'||:old.Ar||'_'||:old.Muszaki_erv);

    END IF;
END;
```

Trigger létrehozása

```
CREATE OR REPLACE TRIGGER név
    BEFORE|AFTER
    INSERT or UPDATE or DELETE
    [of oszlopok] ON táblanév
    [FOR EACH ROW]
    [WHEN feltétel]
BEGIN
    -- PLSQL blokk
END;
```

Automatikus ID generálás

T4	
ID	Adat

```
CREATE SEQUENCE seq1;
```

```
CREATE TABLE T4 (  
  ID int PRIMARY KEY,  
  Adat varchar2(40)  
);
```

```
CREATE OR REPLACE TRIGGER IdGen BEFORE INSERT ON T4 FOR EACH ROW  
BEGIN  
  :new.ID := seq1.nextval;  
END;
```

```
DECLARE  
  newID int;  
BEGIN  
  INSERT INTO T4 (adat) VALUES ('Tulipán') RETURNING ID INTO newID;  
  dbms_output.put_line('Az új rekord azonosítója: '||newID);  
END;
```

```
Az új rekord azonosítója: 1  
1 row(s) inserted.
```

Feladat: kövesse a kódokat, értse meg, és készítsen hasonlót a saját táblájához!

Automatikus érték javítás

T4	
ID	Adat

```
CREATE TABLE T5 (  
  Kod int PRIMARY KEY,  
  Adat varchar2(10),  
  CONSTRAINT adat_ell CHECK (Adat IN ('true', 'false'))  
);
```

```
INSERT INTO T5 VALUES (1, 'igaz');
```

```
ORA-02290: check constraint (WKSP_BITMAN.ADAT_ELL) violated  
1. insert into t5 values (1, 'igaz');
```

Automatikus érték javítás

T5	
Kod	Adat

```
CREATE OR REPLACE TRIGGER T5T1 BEFORE INSERT ON T5 FOR EACH ROW
BEGIN
  IF LOWER(:new.adat) IN ('y','yes','1','t','true','igen','igaz') THEN
    :new.adat := 'true';
  ELSIF LOWER(:new.adat) in ('n','no','0','f','false','nem','hamis') THEN
    :new.adat := 'false';
  END IF;
END;
```

```
INSERT INTO T5 VALUES (1, 'igaz');
```

```
1 row(s) inserted.
```

```
SELECT * FROM T5;
```

KOD	ADAT
1	true

Triggerek kezelése

■ Létrehozott triggerok adatainak listázása:

```
SELECT * FROM user_triggers;
```

TRIGGER_NAME	TRIGGER_TYPE	TRIGGERING_EVENT	TABLE_NAME	STATUS	TRIGGER_BODY
IDGEN	BEFORE EACH ROW	INSERT	T4	ENABLED	BEGIN :new.ID := seq1.nextval; END;
SZINFIGYEL	AFTER EACH ROW	INSERT	AUTO	ENABLED	BEGIN IF :new.szín = 'Piros' THEN insert into AutoTemp values (:new.id); dbms_output.put_line('Tilos színhasználat miatt beszúrás után' 'a rekord törlődik! ID: ' :new.id); END IF; END;
SZINFIGYELTOROL	AFTER STATEMENT	INSERT	AUTO	ENABLED	declare cursor cur is select * from AutoTemp; begin for c_val in cur loop delete from Auto where ID= c_val.ID; end loop; delete from AutoTemp; end;
TRIGBESZUR	AFTER EACH ROW	INSERT	AUTO	ENABLED	BEGIN INSERT INTO AutoNaplo VALUES(sysdate , 'Beszúrás', :new.ID ' ' :new.Rendszam ' ' :new.Tipus ' ' :new.Szin ' ' :new.Kor ' ' :new.Ar ' ' :new.Muszaki_erv); END;
TRIGMODOSIT	AFTER EACH ROW	UPDATE	AUTO	ENABLED	BEGIN INSERT INTO AutoNaplo VALUES(sysdate , 'Módosítás', :new.ID ' ' :new.Rendszam ' ' :new.Tipus ' ' :new.Szin ' ' :new.Kor